
Abstract

An ambitious ethical theory —Alan Gewirth’s ”Principle of Generic Consistency”—
is encoded and analysed in Isabelle/HOL. Gewirth’s theory has stirred much attention in
philosophy and ethics and has been proposed as a potential means to bound the impact
of artificial general intelligence.

1 Introduction

We present an encoding of an ambitious ethical theory —Alan Gewirth’s ”Principle of Generic
Consistency (PGC)”— in Isabelle/HOL. The PGC has stirred much attention in philosophy
and ethics [4] and has been proposed as a potential means to bound the impact of artificial
general intelligence (AGI) [9]. With our contribution we make a first, important step towards
formally assessing the PGC and its potential applications in AI. Our formalisation utilises the
shallow semantical embedding approach [3] and adapts a recent embedding of dyadic deontic
logic in HOL [1] [2].

2 Semantic Embedding of Carmo and Jones’ Dyadic Deontic
Logic (DDL) augmented with Kaplanian contexts

We introduce a modification of the semantic embedding developed by Benzmüller et al. [1]
[2] for the Dyadic Deontic Logic originally presented by Carmo and Jones [5]. We extend this
embedding to a two-dimensional semantics as originally presented by David Kaplan [7] [8].

∗Benzmüller received support from the Volkswagen Foundation (Project CRAP: Consistent Rational Ar-
gumentation in Politics).

2.1 Definition of Types

typedecl w — Type for possible worlds (Kaplan’s ”circumstances of evaluation” or ”counterfactual
situations”)
typedecl e — Type for individuals (entities eligible to become agents)
typedecl c — Type for Kaplanian ”contexts of use”
type-synonym wo = w⇒bool — contents/propositions are identified with their truth-sets
type-synonym cwo = c⇒wo — sentence meaning (Kaplan’s ”character”) is a function from contexts
to contents
type-synonym m = cwo — we use the letter ’m’ for characters (reminiscent of ”meaning”)

2.2 Semantic Characterisation of DDL

2.2.1 Basic Set Operations

abbreviation subset ::wo⇒wo⇒bool (infix v 46 ) where α v β ≡ ∀w . α w −→ β w
abbreviation intersection::wo⇒wo⇒wo (infixr u 48 ) where α u β ≡ λx . α x ∧ β x
abbreviation union::wo⇒wo⇒wo (infixr t 48 ) where α t β ≡ λx . α x ∨ β x
abbreviation complement ::wo⇒wo (∼-[45 ]46 ) where ∼α ≡ λx . ¬α x
abbreviation instantiated ::wo⇒bool (I-[45 ]46 ) where I ϕ ≡ ∃ x . ϕ x
abbreviation setEq ::wo⇒wo⇒bool (infix =s 46 ) where α =s β ≡ ∀ x . α x ←→ β x
abbreviation univSet :: wo (>) where > ≡ λw . True
abbreviation emptySet :: wo (⊥) where ⊥ ≡ λw . False

2.2.2 Set-Theoretic Conditions for DDL

consts
av ::w⇒wo — set of worlds that are open alternatives (aka. actual versions) of w
pv ::w⇒wo — set of worlds that are possible alternatives (aka. potential versions) of w
ob::wo⇒wo⇒bool — set of propositions which are obligatory in a given context (of type wo)

axiomatization where
sem-3a: ∀w . I(av w) and — av is serial: in every situation there is always an open alternative
sem-4a: ∀w . av w v pv w and — open alternatives are possible alternatives
sem-4b: ∀w . pv w w and — pv is reflexive: every situation is a possible alternative to itself
sem-5a: ∀X . ¬(ob X ⊥) and — contradictions cannot be obligatory
sem-5b: ∀X Y Z . (X u Y ) =s (X u Z ) −→ (ob X Y ←→ ob X Z ) and
sem-5c: ∀X Y Z . I(X u Y u Z ) ∧ ob X Y ∧ ob X Z −→ ob X (Y u Z ) and
sem-5d : ∀X Y Z . (Y v X ∧ ob X Y ∧ X v Z ) −→ ob Z ((Z u (∼X )) t Y ) and
sem-5e: ∀X Y Z . Y v X ∧ ob X Z ∧ I(Y u Z ) −→ ob Y Z

lemma True nitpick[satisfy ] oops — model found: axioms are consistent

2.2.3 Verifying Semantic Conditions

lemma sem-5b1 : ob X Y −→ ob X (Y u X ) by (metis (no-types, lifting) sem-5b)
lemma sem-5b2 : (ob X (Y u X ) −→ ob X Y ) by (metis (no-types, lifting) sem-5b)
lemma sem-5ab: ob X Y −→ I(X u Y ) by (metis (full-types) sem-5a sem-5b)
lemma sem-5bd1 : Y v X ∧ ob X Y ∧ X v Z −→ ob Z ((∼X ) t Y ) using sem-5b sem-5d by smt
lemma sem-5bd2 : ob X Y ∧ X v Z −→ ob Z ((Z u (∼X )) t Y ) using sem-5b sem-5d by (smt
sem-5b1 )
lemma sem-5bd3 : ob X Y ∧ X v Z −→ ob Z ((∼X ) t Y ) by (smt sem-5bd2 sem-5b)
lemma sem-5bd4 : ob X Y ∧ X v Z −→ ob Z ((∼X ) t (X u Y )) using sem-5bd3 by auto
lemma sem-5bcd : (ob X Z ∧ ob Y Z ) −→ ob (X t Y ) Z using sem-5b sem-5c sem-5d oops

lemma ob A B ←→ (I(A u B) ∧ (∀X . X v A ∧ I(X u B) −→ ob X B)) using sem-5e sem-5ab
by blast

2.3 (Shallow) Semantic Embedding of DDL

2.3.1 Basic Propositional Logic

abbreviation pand ::m⇒m⇒m (infixr∧ 51 ) where ϕ∧ψ ≡ λc w . (ϕ c w)∧(ψ c w)
abbreviation por ::m⇒m⇒m (infixr∨ 50 ) where ϕ∨ψ ≡ λc w . (ϕ c w)∨(ψ c w)
abbreviation pimp::m⇒m⇒m (infix→ 49 ) where ϕ→ψ ≡ λc w . (ϕ c w)−→(ψ c w)
abbreviation pequ::m⇒m⇒m (infix↔ 48 ) where ϕ↔ψ ≡ λc w . (ϕ c w)←→(ψ c w)
abbreviation pnot ::m⇒m (¬- [52 ]53 ) where ¬ϕ ≡ λc w . ¬(ϕ c w)

2.3.2 Modal Operators

abbreviation cjboxa :: m⇒m (�a- [52 ]53 ) where �aϕ ≡ λc w . ∀ v . (av w) v −→ (ϕ c v)
abbreviation cjdiaa :: m⇒m (♦a- [52 ]53 ) where ♦aϕ ≡ λc w . ∃ v . (av w) v ∧ (ϕ c v)
abbreviation cjboxp :: m⇒m (�p- [52 ]53 ) where �pϕ ≡ λc w . ∀ v . (pv w) v −→ (ϕ c v)
abbreviation cjdiap :: m⇒m (♦p- [52 ]53 ) where ♦pϕ ≡ λc w . ∃ v . (pv w) v ∧ (ϕ c v)
abbreviation cjtaut :: m (>) where > ≡ λc w . True
abbreviation cjcontr :: m (⊥) where ⊥ ≡ λc w . False

2.3.3 Deontic Operators

abbreviation cjod :: m⇒m⇒m (O〈-|-〉54 ) where O〈ϕ|σ〉 ≡ λc w . ob (σ c) (ϕ c)
abbreviation cjoa :: m⇒m (Oa- [53 ]54 ) where Oaϕ ≡ λc w . (ob (av w)) (ϕ c) ∧ (∃ x . (av w) x
∧ ¬(ϕ c x ))
abbreviation cjop :: m⇒m (Oi- [53 ]54 ) where Oiϕ ≡ λc w . (ob (pv w)) (ϕ c) ∧ (∃ x . (pv w) x ∧
¬(ϕ c x ))

2.3.4 Logical Validity (Classical)

abbreviation modvalidctx :: m⇒c⇒bool (b-cM ) where bϕcM ≡ λc. ∀w . ϕ c w — context-dependent
modal validity
abbreviation modvalid :: m⇒bool (b-c) where bϕc ≡ ∀ c. bϕcM c — general modal validity (modally
valid in each context)

2.4 Verifying the Embedding

2.4.1 Avoiding Modal Collapse

lemma bP → OaPc nitpick oops — (actual) deontic modal collapse is countersatisfiable
lemma bP → OiPc nitpick oops — (ideal) deontic modal collapse is countersatisfiable
lemma bP → �aPc nitpick oops — alethic modal collapse is countersatisfiable (implies all other
necessity operators)

2.4.2 Necessitation Rule

lemma NecDDLa: bAc =⇒ b�aAc by simp
lemma NecDDLp: bAc =⇒ b�pAc by simp

2.4.3 Lemmas for Semantic Conditions

abbreviation mboxS5 :: m⇒m (�S5- [52 ]53 ) where �S5ϕ ≡ λc w . ∀ v . ϕ c v
abbreviation mdiaS5 :: m⇒m (♦S5- [52 ]53 ) where ♦S5ϕ ≡ λc w . ∃ v . ϕ c v

lemma C-2 : bO〈A | B〉 → ♦S5(B ∧ A)c by (simp add : sem-5ab)
lemma C-3 : b((♦S5(A ∧ B ∧ C )) ∧ O〈B |A〉 ∧ O〈C |A〉) → O〈(B ∧ C )| A〉c by (simp add : sem-5c)
lemma C-4 : b(�S5(A → B) ∧ ♦S5(A ∧ C ) ∧ O〈C |B〉) → O〈C |A〉c using sem-5e by blast
lemma C-5 : b�S5(A ↔ B) → (O〈C |A〉 → O〈C |B〉)c using C-2 sem-5e by blast
lemma C-6 : b�S5(C → (A ↔ B)) → (O〈A|C 〉 ↔ O〈B |C 〉)c by (metis sem-5b)
lemma C-7 : bO〈B |A〉 → �S5O〈B |A〉c by blast
lemma C-8 : bO〈B |A〉 → O〈A → B | >〉c using sem-5bd4 by presburger

2.4.4 Verifying Axiomatic Characterisation

The following theorems have been taken from the original Carmo and Jones’ paper ([5]
p.293ff).
lemma CJ-3 : b�pA → �aAc by (simp add : sem-4a)
lemma CJ-4 : b¬O〈⊥|A〉c by (simp add : sem-5a)

lemma CJ-5 : b(O〈B |A〉 ∧ O〈C |A〉) → O〈B∧C |A〉c nitpick oops — countermodel found
lemma CJ-5-minus: b♦S5(A ∧ B ∧ C ) ∧ (O〈B |A〉 ∧ O〈C |A〉) → O〈B∧C |A〉c by (simp add :
sem-5c)

lemma CJ-6 : bO〈B |A〉 → O〈B |A∧B〉c by (smt C-2 C-4 )
lemma CJ-7 : bA ↔ Bc −→ bO〈C |A〉 ↔ O〈C |B〉c using sem-5ab sem-5e by blast
lemma CJ-8 : bC → (A ↔ B)c −→ bO〈A|C 〉 ↔ O〈B |C 〉c using C-6 by simp

lemma CJ-9a: b♦pO〈B |A〉 → �pO〈B |A〉c by simp
lemma CJ-9p: b♦aO〈B |A〉 → �aO〈B |A〉c by simp
lemma CJ-9-var-a: bO〈B |A〉 → �aO〈B |A〉c by simp
lemma CJ-9-var-b: bO〈B |A〉 → �pO〈B |A〉c by simp
lemma CJ-10 : b♦p(A ∧ B ∧ C ) ∧ O〈C |B〉 → O〈C |A∧B〉c by (smt C-4 )

lemma CJ-11a: b(OaA ∧ OaB) → Oa(A ∧ B)c nitpick oops — countermodel found
lemma CJ-11a-var : b♦a(A ∧ B) ∧ (OaA ∧ OaB) → Oa(A ∧ B)c using sem-5c by auto

lemma CJ-11p: b(OiA ∧ OiB) → Oi(A ∧ B)c nitpick oops — countermodel found
lemma CJ-11p-var : b♦p(A ∧ B) ∧ (OiA ∧ OiB) → Oi(A ∧ B)c using sem-5c by auto

lemma CJ-12a: b�aA → (¬OaA ∧ ¬Oa(¬A))c using sem-5ab by blast
lemma CJ-12p: b�pA → (¬OiA ∧ ¬Oi(¬A))c using sem-5ab by blast

lemma CJ-13a: b�a(A ↔ B) → (OaA ↔ OaB)c using sem-5b by metis
lemma CJ-13p: b�p(A ↔ B) → (OiA ↔ OiB)c using sem-5b by metis

lemma CJ-O-O : bO〈B |A〉 → O〈A → B |>〉c using sem-5bd4 by presburger

An ideal obligation which is actually possible both to fulfill and to violate entails an actual
obligation ([5] p.319).
lemma CJ-Oi-Oa: b(OiA ∧ ♦aA ∧ ♦a(¬A)) → OaAc using sem-5e sem-4a by blast

Bridge relations between conditional obligations and actual/ideal obligations:

lemma CJ-14a: bO〈B |A〉 ∧ �aA ∧ ♦aB ∧ ♦a¬B → OaBc using sem-5e by blast
lemma CJ-14p: bO〈B |A〉 ∧ �pA ∧ ♦pB ∧ ♦p¬B → OiBc using sem-5e by blast

lemma CJ-15a: b(O〈B |A〉 ∧ ♦a(A ∧ B) ∧ ♦a(A ∧ ¬B)) → Oa(A → B)c using CJ-O-O sem-5e
by fastforce
lemma CJ-15p: b(O〈B |A〉 ∧ ♦p(A ∧ B) ∧ ♦p(A ∧ ¬B)) → Oi(A → B)c using CJ-O-O sem-5e
by fastforce

3 Extending the Carmo and Jones DDL Logical Framework

In the last section, we have modelled Kaplanian contexts by introducing a new type of object
(type c) and modelled sentence meanings as so-called ”characters”, i.e. functions from con-
texts to sets of worlds (type c⇒w⇒o). We also made the corresponding adjustments to the
original semantic embedding of Carmo and Jones’ DDL [1] [2]. So far we haven’t said much
about what these Kaplanian contexts are or which effect they should have on the evaluation
of logical validity. We restricted ourselves to illustrating that their introduction does not have
any influence on the (classical) modal validity of several DDL key theorems. In this section we
introduce an alternative notion of logical validity suited for working with contexts: indexical
validity [7] [8].

3.1 Context Features

Kaplan’s theory (”Logic of Demonstratives” [7]) aims at modelling the behaviour of certain
context-sensitive linguistic expressions like the pronouns ’I’, ’my’, ’you’, ’he’, ’his’, ’she’, ’it’,
the demonstrative pronouns ’that’, ’this’, the adverbs ’here’, ’now’, ’tomorrow’, ’yesterday’,
the adjectives ’actual’, ’present’, and others. Such expressions are known as ”indexicals” and
so Kaplan’s logical system (among others) is usually referred to as a ”logic of indexicals”
(although in his seminal work he referred to it as a ”logic of demonstratives” (LD)) [7]. In
the following we will refer to Kaplan’s logic as the logic ”LD”. It is characteristic of an
indexical that its content varies with context, i.e. they have a context-sensitive character.
Non-indexicals have a fixed character. The same content is invoked in all contexts. Ka-
plan’s logical system models context-sensitivity by representing contexts as tuples of features
(〈Agent(c), Position(c), World(c), Time(c)〉). The agent and the position of context c can
be seen as the actual speaker and place of the utterance respectively, while c’s world and
time stand for the circumstances of evaluation of the expression’s content and allow for the
interaction of indexicals with alethic and tense modalities respectively.

To keep things simple (and relevant for our task) we restrict ourselves to representing a
context c as the pair: 〈Agent(c), World(c)〉. For this purpose we represent the functional
concepts ”Agent” and ”World” as logical constants.

consts Agent ::c⇒e — function retrieving the agent corresponding to context c
consts World ::c⇒w — function retrieving the world corresponding to context c

3.2 Logical Validity

Kaplan’s notion of (context-dependent) logical truth for a sentence corresponds to its (context-
sensitive) formula (of type c⇒w⇒bool i.e. m) being true in the given context and at its

corresponding world.

abbreviation ldtruectx ::m⇒c⇒bool (b-c-) where bϕcc ≡ ϕ c (World c) — truth in the given context

Kaplan’s LD notion of logical validity for a sentence corresponds to its being true in all
contexts. This notion is also known as indexical validity.

abbreviation ldvalid ::m⇒bool (b-cD) where bϕcD ≡ ∀ c. bϕcc — LD validity (true in every context)

Here we show that indexical validity is indeed weaker than its classical modal counterpart
(truth at all worlds for all contexts):

lemma bAc =⇒ bAcD by simp
lemma bAcD =⇒ bAc nitpick oops — countermodel found

Here we show that the interplay between indexical validity and the DDL modal and deontic
operators does not result in modal collapse.

lemma bP → OaPcD nitpick oops
lemma bP → �aPcD nitpick oops

Next we show that the necessitation rule does not work for indexical validity (in contrast to
classical modal validity as defined for DDL).

lemma NecLDa: bAcD =⇒ b�aAcD nitpick oops
lemma NecLDp: bAcD =⇒ b�pAcD nitpick oops

The following can be seen as a kind of ’analytic/a priori necessity’ operator (to be contrasted
to the more traditional metaphysic necessity). In Kaplan’s framework, a sentence being
logically (i.e. indexically) valid means its being true a priori : it is guaranteed to be true in
every possible context in which it is uttered, even though it may express distinct propositions
in different contexts. This correlation between indexical validity and a prioricity has also
been claimed in other two-dimensional semantic frameworks [10].

abbreviation ldvalidbox :: m⇒m (�D- [52 ]53 ) where �Dϕ ≡ λc w . bϕcD — notice the D superscript
lemma b�DϕcC ≡ ∀ c.bϕcc by simp — this operator works analogously to the box operator in modal
logic S5

Quite trivially, the necessitation rule works for the combination of indexical validity with the
previous operator.

lemma NecLDa: bAcD =⇒ b�DAcD by simp
lemma NecLDp: bAcD =⇒ b�DAcD by simp

The operator above is not part of the original Kaplan’s LD ([7]) and has been added by us in
order to better highlight some semantic features of our formalisation of Gewirth’s argument
in the next section and to being able to use the necessitation rule for some inference steps.

3.3 Quantification

We also enrich our logic with (higher-order) quantifiers (using parameterised types).

abbreviation mforall ::( ′t⇒m)⇒m (∀ ) where ∀ Φ ≡ λc w .∀ x . (Φ x c w)
abbreviation mexists::( ′t⇒m)⇒m (∃ ) where ∃ Φ ≡ λc w .∃ x . (Φ x c w)
abbreviation mforallBinder ::( ′t⇒m)⇒m (binder∀ [8 ]9 ) where ∀ x . (ϕ x ) ≡ ∀ϕ
abbreviation mexistsBinder ::( ′t⇒m)⇒m (binder∃ [8 ]9 ) where ∃ x . (ϕ x ) ≡ ∃ϕ

Before starting our formalisation in the next section. We show that the axioms defined so
far are consistent. Rather surprisingly, the nunchaku model finder states that no model has
been found, while nitpick is indeed able to find one:

lemma True nunchaku[satisfy ] nitpick[satisfy ] oops

4 Gewith’s Argument for the Principle of Generic Consistency
(PGC)

Alan Gewirth’s meta-ethical position is known as moral (or ethical) rationalism. According
to it, moral principles are knowable a priori, by reason alone. Immanuel Kant is perhaps
the most famous figure who has defended such a position. He has argued for the existence
of upper moral principles (e.g. his ”categorical imperative”) from which we can reason (in a
top-down fashion) in order to deduce and evaluate other more concrete maxims and actions.
In contrast to Kant, Gewirth attempts to derive such upper moral principles by starting from
non-moral considerations alone, namely from an agent’s self-reflection. Gewirth’s Principle of
Generic Consistency (PGC) asserts that any agent (by virtue of its self-understanding as an
agent) is rationally committed to asserting that (i) it has rights to freedom and well-being,
and (ii) that all other agents have those same rights. Gewirth claims that, in his informal
proof, the latter generalisation step (from ”I” to all individuals) is done on purely logical
grounds and does not presuppose any kind of universal moral principle. Gewirth’s result is
thus meant to hold with some kind of apodicticity (i.e. necessity). Deryck Beyleveld, author
of an authoritative book on Gewirth’s argument, puts it this way: ”The argument purports
to establish the PGC as a rationally necessary proposition with an apodictic status for any
PPA equivalent to that enjoyed by the logical principle of noncontradiction itself.” ([4] p. 1)
If this is correct, then he succeeded in the task that Kant set himself, i.e. to found certain
basic principles of morality in reason alone.

The argument for the PGC employs what Gewirth calls ”the dialectically necessary method”
within the ”internal viewpoint” (perspective) of an agent. Although the drawn inferences
are relative to the reasoning agent, Gewirth further argues that ”the dialectically necessary
method propounds the contents of this relativity as necessary ones, since the statements it
presents reflect judgements all agents necessarily make on the basis of what is necessarily
involved in their actions ... The statements the method attributes to the agent are set
forth as necessary ones in that they reflect what is conceptually necessary to being an agent
who voluntarily or freely acts for purposes he wants to attain.” ([6]). In other words, the
”dialectical necessity” of the assertions and inferences made in the argument comes from the
definitional features (conceptual analysis) of the involved notions of agency, purposeful action,
obligation, rights, etc. Hence the alternative notions of logical (i.e. indexical) validity and ’a
priori necessity’, developed in Kaplan’s logical framework LD, have been considered by us as
appropriate to model this kind of ”dialectical necessity”.

4.1 Conceptual Explications

type-synonym p = e⇒m — Type for properties (function from individuals to sentence meanings)

4.1.1 Agency

The type chosen to represent what Gewirth calls ”purposes” is not essential for the argument’s
validity. We choose to give ”purposes” the same type as sentence meanings (type ’m’),
so ”acting on a purpose” would be represented in an analogous way to having a certain
propositional attitude (e.g. ”desiring that some proposition obtains”).

consts ActsOnPurpose:: e⇒m⇒m — ActsOnPurpose(A,E) gives the meaning of the sentence ”A is
acting on purpose E”
consts NeedsForPurpose:: e⇒p⇒m⇒m — NeedsForPurpose(A,P,E) gives the meaning of ”A needs
to have property P in order to reach purpose E”

In Gewirth’s argument, an individual with agency (i.e. capable of purposive action) is said to
be a PPA (prospective purposive agent).

definition PPA:: p where PPA a ≡ ∃ E . ActsOnPurpose a E — Definition of PPA

We have added the following axiom in order to guarantee the argument’s logical correctness.
It basically says that being a PPA is identity-constitutive for an individual (i.e. it’s an essential
property).

axiomatization where essentialPPA: b∀ a. PPA a → �D(PPA a)cD — being a PPA is an essential
property

Quite interestingly, the axiom above entails, as a corollary, a kind of ability for a PPA to
recognise other PPAs. For instance, if some individual holds itself as a PPA (i.e. seen from
its own perspective/context ’d’) then this individual (Agent(d)) is considered as a PPA from
any other agent’s perspective/context ’c’.

lemma recognizeOtherPPA: ∀ c d . bPPA (Agent d)cd −→ bPPA (Agent d)cc using essentialPPA by
blast

4.1.2 Goodness

Gewirth’s concept of (subjective) goodness, as employed in his argument, applies to purposes
and is relative to some agent. It is therefore modelled as a binary relation relating an individ-
ual (type ’e’) with a purpose (type ’m’). Other readings given by Gewirth’s for the expression
”P is good for A” include among others: ”A attaches a positive value to P”, ”A values P
proactively” and ”A is motivated to achieve P”.

consts Good ::e⇒m⇒m

The following axioms interrelate the concept of goodness with the concept of agency, thus
providing the above concepts with some meaning (by framing their inferential roles). Notice
that such meaning-constitutive axioms (which we call ”explications”) are given as indexically
valid (i.e. a priori) sentences.

axiomatization where explicationGoodness1 : b∀ a P . ActsOnPurpose a P → Good a PcD
axiomatization where explicationGoodness2 : b∀ P M a. Good a P ∧ NeedsForPurpose a M P →
Good a (M a)cD
axiomatization where explicationGoodness3 : b∀ϕ a. ♦pϕ → O〈ϕ | �DGood a ϕ〉cD

Below we show that all axioms defined so far are consistent:

lemma True nitpick[satisfy , card c = 1 , card e = 1 , card w = 1 ] oops — one-world model found
(card w=1)

The first two assertions above have been explicitly provided by Gewirth as premises of his
argument. The third axiom, however, has been added by us as an implicit premise in order
to render Gewirth’s proof as correct. This axiom aims at representing the intuitive notion of
”seeking the good”. In particular, it asserts that, from the point of view of an agent, necessar-
ily good purposes are not only action motivating, but also entail an instrumental obligation
to their realisation. The notion of necessity here involved is not the usual metaphysical one
(which is represented in DDL with the modal box operator �a), but the linguistic one intro-
duced above (�D) derived from indexical validity, signaling that an agent holds some purpose
as being true almost ’by definition’ (i.e. a priori). This sets quite high standards for the
kind of purposes an agent would ever take to be (instrumentally) obligatory and is indeed the
weakest implicit premise we could come up with so far (taking away the �D ’a priori necessity’
operator would indeed make this premise much stronger and our proof less credible).

4.1.3 Freedom and Well-Being

According to Gewirth, enjoying freedom and well-being (which we take together as a predicate:
FWB) is the property which represents the ”necessary conditions” or ”generic features” of
agency (i.e. being capable of purposeful action). Gewirth argues, the property of enjoying
freedom and well-being (FWB) is special amongst other action-enabling properties, in that
it is always required in order to act on any purpose (no matter which one). As such we
can reasonably demand that FWB be (metaphysically) possible for every agent. As before,
we take this demand to be an a priori characteristic of the concept of FWB and therefore
axiomatise it as an indexically valid sentence.

consts FWB ::p — Enjoying freedom and well-being (FWB) is a property (i.e. has type e⇒m)

axiomatization where
explicationFWB1 : b∀ P a. NeedsForPurpose a FWB PcD

We use model finder nitpick to verify that all axioms defined so far are consistent. Nitpick
can indeed find a ’small’ model with cardinality one for the sets of worlds and contexts.

lemma True nitpick[satisfy , card c = 1 , card e = 1 , card w = 1 ] oops — one-world model found

At some point in Gewirth’s argument we have to show that there exists an (instrumental) obli-
gation to enjoying freedom and well-being (FWB). Since, according to the so-called ”Kant’s
law” (which is a corollary of DDL), impossible or necessary things cannot be obligatory, we
add the following as an a priori characteristic of FWB: The state of affairs of some individual
(e.g. ”I”) enjoying freedom and well-being is contingent.

axiomatization where explicationFWB2 : b∀ a. ♦p FWB acD
axiomatization where explicationFWB3 : b∀ a. ♦p ¬FWB acD

As a result of enforcing the contingency of FWB, the models found by nitpick now have a
cardinality of two for the set of worlds:

lemma True nitpick[satisfy , card c = 1 , card e = 1 , card w = 1 , expect=none] oops — no model
found for one-world models
lemma True nitpick[satisfy , card c = 1 , card e = 1 , card w = 2 ] oops — models need now at least
two worlds

4.1.4 Obligation and Interference

Kant’s Law (”ought implies can”) is derivable directly from DDL: If ϕ oughts to obtain
then ϕ is possible. Note that we will use for the formalisation of Gewirth’s argument the
DDL ideal obligation operator (Oi) but we could have also used (mutatis mutandis) the DDL
actual obligation operator (Oa).

lemma bOiϕ → ♦pϕc using sem-5ab by simp

Furthermore, we have seen the need to postulate the following (implicit) premise in order
to validate the argument. This axiom can be seen as a variation of the so-called Kant’s
law (”ought implies can”), i.e. an impossible act cannot be obligatory. In the same vein,
our variation can be read as ”ought implies ought to can” and is closer to Gewirth’s own
description: that having an obligation to do X implies that ”I ought (in the same sense and
the same criterion) to be free to do X, that I ought not to be prevented from doing X, that
my capacity to do X ought not to be interfered with.” ([6] p. 91-95)

axiomatization where OIOAC : bOiϕ → Oi(♦aϕ)cD

Concerning the concept of interference, we state that the existence of an individual (success-
fully) interfering with some state of affairs S implies that S cannot possibly obtain in any
of the actually possible situations (and the other way round). Note that for this definition
we have employed a possibility operator (♦a) which is weaker than metaphysical possibility
(♦a) (see Carmo and Jones DDL framework [5] for details). Also note that we have also em-
ployed the (stronger) classical notion of modal validity instead of indexical validity. (So far
we haven’t been able to get theorem provers and model finders to prove/disprove Gewirth’s
proof if formalizing this axiom as simply indexically valid.)

consts InterferesWith::e⇒m⇒m — an individual can interfere with some state of affairs (from
obtaining)
axiomatization where explicationInterference: b(∃ b. InterferesWith b ϕ) ↔ ¬♦aϕc

From the previous axiom we can prove following corollaries: If someone (successfully) interferes
with agent ’a’ having FWB, then ’a’ can no longer possibly enjoy its FWB (and the other
way round).

lemma b∀ a. (∃ b. InterferesWith b (FWB a)) ↔ ¬♦a(FWB a)c using explicationInterference by
blast
lemma InterferenceWithFWB : b∀ a. ♦a(FWB a) ↔ (∀ b. ¬InterferesWith b (FWB a))c using
explicationInterference by blast

4.1.5 Rights and Other-Directed Obligations

Gewirth points out the existence of a correlation between an agent’s own claim rights and
other-referring obligations (see e.g. [6], p. 66). A claim right is a right which entails duties
or obligations on other agents regarding the right-holder (so-called Hohfeldian claim rights
in legal theory). We model this concept of claim rights in such a way that an individual
’a’ has a (claim) right to some property ’P’ if and only if it is obligatory that every (other)
individual ’b’ does not interfere with the state of affairs ’P(a)’ from obtaining. Since there is no
particular individual to whom this directive is addressed, this obligation has been referred to
by Gewirth as being ”other-directed” (aka. ”other-referring”) in contrast to ”other-directing”

obligations which entail a moral obligation for some particular subject ([4] p. 41,51). This
latter distinction is essential to Gewirth’s argument.

definition RightTo::e⇒(e⇒m)⇒m where RightTo a ϕ ≡ Oi(∀ b. ¬InterferesWith b (ϕ a))

Now that all needed axioms and definitions are in place, we use model finder nitpick to show
that they are consistent:

lemma True nitpick[satisfy , card c = 1 , card e = 1 , card w = 2 ] oops — models with at least two
worlds found

4.2 Formal Proof of Gewirth’s PGC

Following Beyleveld’s summary ([4], ch. 2), the main steps of the argument are (with original
numbering):

(1) I act voluntarily for some (freely chosen) purpose E (equivalent –by definition– to: I am
a PPA).

(2) E is (subjectively) good (i.e. I value E proactively).

(3) My freedom and well-being (FWB) are generically necessary conditions of my agency
(i.e. I need them to achieve any purpose whatsoever).

(4) My FWB are necessary goods (at least for me).

(5) I have (maybe nobody else does) a claim right to my FWB.

(13) Every PPA has a claim right to their FWB.

In the following we present a formalised proof for the Principle of Generic Consistency (PGC):
”Every PPA has a claim right to its freedom and well-being”.

theorem PGC : shows ∀C . bPPA (Agent C ) → (RightTo (Agent C ) FWB)cC
proof − {

fix C ::c — ’C’ is some arbitrarily chosen context (agent’s perspective)
let ?I = (Agent C ) — ’I’ is/am the agent with perspective ’C’
{

fix E ::m — ’E’ is some arbitrarily chosen purpose
{

assume P1 : bActsOnPurpose ?I EcC — (1) I act voluntarily on purpose E
from P1 have P1a: bPPA ?I cC using PPA-def by auto — (1a) I am a PPA
from P1 have C2 : bGood ?I EcC using explicationGoodness1 essentialPPA by meson — (2)

purpose E is good for me
from explicationFWB1 have C3 : b∀ P . NeedsForPurpose ?I FWB PcD by simp — (3) I need

FWB for any purpose whatsoever
hence ∃P .bGood ?I P ∧ NeedsForPurpose ?I FWB PcD using explicationFWB2 explication-

Goodness3 sem-5ab by blast
hence bGood ?I (FWB ?I )cD using explicationGoodness2 by blast — FWB is (a priori) good

for me (in a kind of definitional sense)
hence C4 : b�D(Good ?I (FWB ?I ))cC by simp — (4) FWB is an (a priori) necessary good for

me
have bO〈FWB ?I | �D(Good ?I ) (FWB ?I )〉cC using explicationGoodness3 explicationFWB2

by blast — I ought to pursue my FWB on the condition that I consider it to be a necessary good

hence bOi(FWB ?I )cC using explicationFWB2 explicationFWB3 C4 CJ-14p by fastforce —
There is an (other-directed) obligation to my FWB

hence bOi(♦a(FWB ?I ))cC using OIOAC by simp — It must therefore be the case that my
FWB is possible

hence bOi(∀ a. ¬InterferesWith a (FWB ?I ))cC using InterferenceWithFWB by simp — There
is an obligation for others not to interfere with my FWB

hence C5 : bRightTo ?I FWBcC using RightTo-def by simp — (5) I have a claim right to my
freedom and well-being

}
hence bActsOnPurpose ?I E → RightTo ?I FWBcC by (rule impI ) — I have a claim right to my

freedom and well-being (since I act on some purpose E)
}
hence b∀ P . ActsOnPurpose ?I P → RightTo ?I FWBcC by (rule allI ) — ”allI” is a logical

generalisation rule: ”all-quantifier introduction”
hence bPPA ?I → RightTo ?I FWBcC using PPA-def by simp — (seen from my perspective C)

I have a claim right to my freedom and well-being since I am a PPA
hence bPPA (Agent C ) → RightTo (Agent C ) FWBcC by simp — (seen from the perspective C)

C’s agent has a claim right to its freedom and well-being since it is a PPA
}

thus C13 : ∀C . bPPA (Agent C ) → (RightTo (Agent C ) FWB)cC by (rule allI ) — (13) For every
perspective C: C’s agent has a claim right to its freedom and well-being
qed

Regarding the last inference step, given that the context (agent’s perspective) ’C’ has been
arbitrarily fixed at the beginning, we can use again the ”all-quantifier introduction” rule to
generalise the previous assertion to all possible contexts ’C’ (and agents ’Agent(C)’). Note
that the generalisation from ”I” to all individuals has been done on purely logical grounds and
does not involve any kind of universal moral principle. This is a main requirement Gewirth
has set for his argument.

